


Symbolic SPICE



Circuit Analyzer and Approximator

Application Note

AN-008:

Gain-Bandwidth-Product Effects in a
MFB Band-Pass Active Filter

by Gregory M. Wierzba

A) Introduction

Op-amps look nearly ideal in many audio applications because the gain of the amplifier in the audio band is very large. Active filters can realize low-pass, high-pass, band-pass and band-stop transfer functions. Higher order active filters are usually made by cascading second order building blocks. The denominator of a second order filter is of the form $s^2 + (\omega_o/Q_o)s + \omega_o^2$. In many audio applications, low-pass and high-pass filters have a Q_o that is usually less than one but in band-pass and band-stop filters it may be much larger than one. All active filters have a maximum value for their $\omega_o Q_o$ product due to the op-amps gain-bandwidth-product (GBP). If the filter operates much below this product then the filter will look nearly ideal. The GBP is the product of the op-amp's low frequency open-loop gain times its -3 dB bandwidth. This application note shows how Symbolic SPICE® can be used to evaluate and design an RC active band-pass filter so as to reduce the effects of GBP.

The schematic shown in Fig. 1 is that of a Multiple-Feedback (MFB) Band-Pass Filter [1-2]. One application for this type of filter is in the design of a Graphic Equalizer [3-5].

A Symbolic SPICE input file, **mfb_bp.cir**, is given in Table 1. This file uses Symbolic SPICE's definition [6] of an ideal op-amp which begins with the letters XOA.




Figure 1. Multiple-feedback band-pass active filter

Table 1. Symbolic SPICE input file **mfb_bp.cir**

```
MFB BAND-PASS FILTER
V1 1 0 AC 1
R1 1 3
R2 3 0
R5 4 2
C3 3 4
C4 3 2
XOA 0 4 2
.END
```

B) Running Symbolic SPICE

Running [6] the input file shown in Table 1, the following are the prompts. The user responses are shown in bold:

```
Symbolic SPICE - Circuit Analyzer and Approximator
Demo Version 3.1
(C) Copyright 2010 by Willow Electronics, Inc.

INPUT FILE NAME [.cir] : mfb_bp
OUTPUT FILENAME [mfb_bp.det] : (hit enter)
Determinant string sorted according to orders of some variable? (y/n) : n
Numerical evaluation of the results? (y/n) : n
Discard terms if their magnitude falls below a threshold? (y/n) : n
Check and solve for second order filter functions? (y/n) : y
FILTER FUNCTION FILE NAME [mfb_bp.fun] : (hit enter)
Solve for a variable or expression? (y/n) : y
Available Unknowns:
V1 V2 V3
*Ignore nodes 5 and higher if present. They are used for internal numbering.
Valid Operators: +, -, *, /, ( ), { }, [ ]
Equation: v2
Solve for another variable or expression? (y/n) : n
```

C) Symbolic SPICE Determinant Output

The output file **mfb_bp.det** listed in Table 2 is the matrix written by Symbolic SPICE and the transfer function requested.

Table 2. Symbolic SPICE Output File **mfb_bp.det**

```
MFB BAND-PASS FILTER

[0 ] [-G1          -sC4          sC4+sC3+G2+G1      ] [V1 ]
[0 ]=[0          -G5          -sC3          ] [V2 ]
[1 ] [1          0           0           ] [V3 ]

*Ignore nodes 5 and higher if present. They are used for internal numbering.

Numerator of: v2

TERMS SORTED ACCORDING TO POWERS OF s

s**1 terms:

- sC3*G1
```

```

*****
Denominator of: v2
TERMS SORTED ACCORDING TO POWERS OF s
s**2 terms:
+ sC4*sC3
s**1 terms:
+ sC4*G5 + sC3*G5
s**0 terms:
+ G5*G2 + G5*G1
*****

```

D) Symbolic SPICE Filter Function Output

Symbolic SPICE will examine any transfer function to determine if its denominator is second order in the Laplace operator s . If this is the case then it finds the parameters associated with the different filter functions. Symbolic SPICE finds formulas for Q_o , ω_o and H_i [6].

The output file **mfb_bp.fun** listed in Table 3 contains the filter parameters solved for by Symbolic SPICE.

Table 3. Symbolic SPICE Output File **mfb_bp.fun**

```

MFB BAND-PASS FILTER
SECOND ORDER FILTER PARAMETERS:
Qo is:
Sqrt{( + C4*C3)*( + G2 + G1)}
-----
( + C4 + C3)*Sqrt{ + G5}
Wo**2 is:
( + G5)*( + G2 + G1)
-----
( + C4*C3)
*****
There exists a BAND PASS filter for : v2
BAND PASS GAIN (Hbp) is:
( - C3*G1)
-----
( + G5)*( + C4 + C3)
*****

```

Symbolic SPICE's format is a collection of strings of symbols. This is usually not how most people view equations, so you may need to do some minor editing. Thus we have:

$$Q_o = \frac{\sqrt{C_4 C_3}}{C_4 + C_3} \sqrt{\frac{G_2 + G_1}{G_5}} = \frac{\sqrt{C_4 C_3}}{C_4 + C_3} \sqrt{\frac{R_5}{R_2 \parallel R_1}} \quad (1)$$

$$\omega_o = \sqrt{\frac{G_5(G_2 + G_1)}{C_4 C_3}} = \sqrt{\frac{1}{C_4 C_3 R_5 (R_2 \parallel R_1)}} \quad (2)$$

$$H_{bp} = -\frac{C_3 G_1}{G_5 (C_4 + C_3)} = -\frac{C_3 R_5}{R_1 (C_4 + C_3)} \quad (3)$$

Another useful formula is that of the -3 dB bandwidth (radians/sec)

$$\text{Bandwidth} = \frac{\omega_o}{Q_o} = \frac{G_5 (C_4 + C_3)}{C_4 C_3} = \frac{(C_4 + C_3)}{R_5 C_4 C_3} \quad (4)$$

E) Design Procedure

Given a multiple-feedback band-pass filter, we can measure the component values or use the nominal values and predict the filter parameters of Q_o , ω_o and H_{bp} with the equations above.

Reversing this process would allow the selection of components given the filter parameters. Since there are three filter parameters (Q_o , ω_o and H_{bp}) and five components (R_1 , R_2 , C_3 , C_4 and R_5) then we can pick two and solve for the rest. Note that this is one way out of many to pick the components.

Eqns.1-4 [1] have been used to develop a design procedure:

- 1) Pick $C_3 = C_4$ to be a standard value
- 2) From Eqn. 4, $R_5 = 2 Q_o / (\omega_o C_3)$
- 3) From Eqn. 3, $R_1 = Q_o / (H_{bp} \omega_o C_3)$
- 4) From Eqn. 1, $R_2 = Q_o / [(2 Q_o^2 - H_{bp}) \omega_o C_3]$

Note that the last step implies that $2 Q_o^2 > H_{bp}$. If this is not true then develop a new design procedure with $C_3 \neq C_4$.

F) Design Example

The merits of constant Q_o filters for various Graphic Equalizers are discussed in [7]. In [8], a 1-octave equalizer is designed using the design procedure in Section E with $Q_o = 2$ and $H_{bp} = 4$. To expand this to an ultimate 1/3-octave equalizer with 32 slide pots, a constant Q_o of 4.318 is recommended in [7]. If we use the suggested value of $H_{bp} = 4$ then we will have a maximum gain of 12 dB.

Although we could pick the entire 32 band pass filters using these specifications from $f_o = 16$ Hz to 20k Hz in 1/3-octave steps, let's leave that for another application note and just look at one filter to show a potential problem due to the gain-bandwidth-product of the op-amp.

For $Q_o = 4.318$, $f_o = 20$ k Hz and $H_{bp} = 4$ then

- 1) Pick $C_3 = C_4 = 390$ p F
- 2) $R_5 = 2 Q_o / (\omega_o C_3) = 176.21$ k Ω (Pick 180k Ω)
- 3) $R_1 = Q_o / (H_{bp} \omega_o C_3) = 22.027$ k Ω (Pick 22k Ω)
- 4) $R_2 = Q_o / [(2 Q_o^2 - H_{bp}) \omega_o C_3] = 2.6466$ k Ω (Pick 2.7k Ω)

Using Eqns. 1-3, we can re-calculate the filter parameters with the nearest standard components or we could just re-run Symbolic SPICE. The revised input file **mfb_bp_std.cir** is listed in Table 4.

Table 4. Symbolic SPICE input file **mfb_bp_std.cir**

```
MFB BAND-PASS FILTER
V1 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XOA 0 4 2
.END
```

The output file **mfb_bp_std.fun** listed in Table 5 contains the filter parameters solved for by Symbolic SPICE.

Table 5. Symbolic SPICE Output File **mfb_bp.fun**

```

MFB BAND-PASS FILTER

SECOND ORDER FILTER PARAMETERS:

Qo is:

Sqrt{( + C4*C3)*( + G2 + G1)}
-----
( + C4 + C3)*Sqrt{ + G5}

= 4.32575

Wo**2 is:

( + G5)*( + G2 + G1)
-----
( + C4*C3)

fo = 19614.4Hz

*****

There exists a BAND PASS filter for : v2

BAND PASS GAIN (Hbp) is:

( - C3*G1)
-----
( + G5)*( + C4 + C3)

= -4.0909

*****

```

F) Symbolic SPICE's Non-ideal Op-Amp

Symbolic SPICE has a model for a non-ideal op-amp [3]. This is shown in Fig. 2 where the output is V_{in} times $(2\pi \cdot \text{gain-bandwidth-product} / s)$.




Figure 2. Non-ideal op-amp

The element definition for this model is :

```
XNOA[name] <(+)input> <(-)input> <output> {ignore rest of line}
```

This is the format of a subcircuit model where for a numeric program the subcircuit model begins with `.SUBCKT` and ends with `.ENDS`.

Symbolic SPICE ignores these subcircuit lines of code so that it can use its own definition of gain-bandwidth-product.

When this element is used in the input file, a new set of menu options appears. This is because this element invokes an approximation algorithm [9] which is only valid for second order active filters. If the user wishes to create an expression for the GBP, it must be done with an equivalent circuit but not as a subcircuit.

Table 6 lists an edited version of the file **mfb_bp_std.cir** where XOA was replaced with XNOA.

Table 6. Symbolic SPICE input file **mfb_bp_noa.cir**

```
MFB BAND-PASS FILTER
V1 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XNOA 0 4 2
.END
```

Suppose we evaluate this filter using a low cost op-amp like a uA741 which has a gain-bandwidth-product of about 1M Hz. Running the input file shown in Table 6, the following are the prompts. The user responses are shown in bold:

Symbolic SPICE - Circuit Analyzer and Approximator
 Demo Version 3.1
 (C) Copyright 2010 by Willow Electronics, Inc.

```
INPUT FILE NAME [.cir] : mfb_bp_noa
Numerical evaluation of the results? (y/n) : y
Discard terms if their magnitude falls below a threshold? (y/n) : n
ERROR ANALYSIS FILE NAME [mfb_bp_noa.err] : (hit enter)
What is the GBP (in Hz) of XNOA? (y/n) : 1meg
```

The output file **mfb_bp_noa.err** listed in Table 7 is the analysis done by Symbolic SPICE on the shifting of poles in the characteristic equation. The approximation algorithm finds the change in f_o and Q_o symbolically due to the elements in the circuit, the gain-bandwidth-product and the symbolic values of f_o and Q_o .

Table 7. Symbolic SPICE Output File **mfb_bp_noa.err**

MFB BAND-PASS FILTER

```
*****
* Qo and fo *
*****
```

Qo is:

```
SQRT{( + C4*C3)*( + G2 + G1)}
-----
( + C4 + C3)*SQRT{ + G5}

= 4.32575
```

(2*PI*fo)**2 is:

```
( + G5)*( + G2 + G1)
-----
( + C4*C3)
```

fo = 19614.4 Hz

```
*****
* DQo/Qo = (D2-D0)foQo - Dfo/fo *
*****
```

D2-D0: where k{i} = 1/GBP{i}

The numerator is:

```
+0 TIMES

+ 1
```

The denominator is:

```
( + G5*C4*C3)*( + G2 + G1)

*****
* Dfo/fo = (D2-D1)fo/(2Qo) *
*****
```

D2-D1 : where k{i} = 1/GBP{i}

The numerator is:

```
-C3*k TIMES

+ G2 + G1
```

The denominator is:

```
( + G5)*( + C4 + C3)

*****
*       NUMERICAL EVALUATION       *
*****
```

Dfo/fo = -0.0848468

DQo/Qo = 0.0848468

Thus we have that:

$$\frac{\Delta f_o}{f_o} \approx -\frac{C_3(G_2 + G_1)}{G_5(C_4 + C_3)} \frac{f_o}{2Q_o} \frac{1}{\text{GBP}} = -0.0848468 \quad (5)$$

$$\frac{\Delta Q_o}{Q_o} \approx \frac{C_3(G_2 + G_1)}{G_5(C_4 + C_3)} \frac{f_o}{2Q_o} \frac{1}{\text{GBP}} = 0.0848468 \quad (6)$$

What this means symbolically is

$$\text{Actual } f_o \triangleq \hat{f}_o \approx f_o \left(1 + \frac{\Delta f_o}{f_o} \right) \quad (7)$$

$$\text{Actual } Q_o \triangleq \hat{Q}_o \approx Q_o \left(1 + \frac{\Delta Q_o}{Q_o} \right) \quad (8)$$

What these results mean numerically is that there is a *decrease* in the center frequency f_o of approximately 8.48468 % and an *increase* in the filter Q_o of approximately 8.48468 %. In other words, the effect of having a 1M Hz GBP is to shift f_o from 19,614.4 to an actual value of 19,614.4 (1 - 0.0848468) = 17,950 Hz and to shift Q_o from 4.32575 to an actual value of 4.32575 (1 + 0.0848468) = 4.6928.

The shifting in f_o and Q_o shifts the bandwidth which is f_o / Q_o . In this case the bandwidth was shifted from 19,614.4 / 4.32575 = 4,534.3 Hz to an actual value of 17,950 / 4.6928 = 3,825 Hz.

Therefore our filter is covering a different band of frequencies than we designed for.

G) PSpice Evaluation

PSpice allows for controlled sources with Laplace expressions. A one-pole model is of the form:

$$A_o \frac{2\pi f_o}{s + 2\pi f_o}$$

where A_o is the op-amp open-loop gain at low frequencies, f_o is the op-amp's open-loop gain -3 dB frequency and $A_o f_o$ is the op-amp's gain-bandwidth-product.

Let's verify the Symbolic SPICE results with PSpice using an ideal op-amp in the first simulation and a one-pole model for the op-amp in the second. The input file is listed in Table 8 where $A_o = 200k$, $f_o = 5$ Hz and the GBP = 1M Hz. The output is graphed in Fig. 3.

Table 8. PSpice input file **bandpass.cir**

```
BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XOA 0 4 2 IDEAL
.SUBCKT IDEAL 1 2 3
RI 1 2 1000MEG
EA 3 0 1 2 1000MEG
.ENDS IDEAL
.AC DEC 20000 2K 200K
.PROBE
.END

BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XNOA 0 4 2 uA741
.SUBCKT uA741 1 2 3
EA 3 0 LAPLACE {V(1,2)} =
+ {(200K)*(6.283*5)/(S + (6.283*5))}
.ENDS uA741
.AC DEC 20000 2K 200K
.PROBE
.END
```

From Fig. 3, for the ideal op-amp the peak frequency is 19.615k Hz, the peak gain is 4.0909 and the -3 dB bandwidth is 22.002k - 17.486k = 4.516k Hz. Thus

$$f_o = 19.615\text{k Hz} \quad H_o = 4.0909 \quad Q_o = 19.615\text{k} / 4.515\text{k} = 4.3434$$

and this closely agrees with Symbolic SPICE.

From Fig. 3, for a non-ideal op-amp the peak frequency is 18.131k Hz, the peak gain is 4.0398 and the -3 dB bandwidth is 20.201k - 16.274k = 3.927k Hz. Thus

$$\hat{f}_o = 18.131\text{k Hz} \quad \hat{H}_o = 4.0398 \quad \hat{Q}_o = 18.131\text{k} / 3.927\text{k} = 4.617$$

Comparing to Symbolic SPICE, this is a difference of +1% for f_o and -1.6% for Q_o which is close given that the algorithm used is always an approximation.




Figure 3. Effect of GBP on the voltage gain

H) Redesign

In order to reduce the error due to the GBP of the op-amp suppose that we revisit Eqns. 5 and 6. Since both the error in f_o and Q_o are inversely proportional to the GBP, we could simply pick an op-amp with a higher GBP. For example, an NE5534 op-amp has a GBP of 10M Hz. This would lower the error by a factor of 10, that is, there would be a *decrease* in the center frequency f_o of approximately 0.848468 % and an *increase* in the filter Q_o of 0.848468 %. In other words, the effect of having a 10M Hz GBP would be to shift f_o from 19,614.4 to an actual value of $19,614.4 (1 - 0.00848468) = 19,448$ Hz and to shift Q_o from 4.32575 to an actual value of $4.32575 (1 + 0.00848468) = 4.3625$. Thus the bandwidth would shifted from $19,614.4 / 4.32575 = 4,534.3$ Hz to an actual value of $19,448 / 4.3625 = 4,458$ Hz.

Another improvement could also be found by re-examining Eqns. 5 and 6 and substituting in Eqn. 1. After a little algebra we can show that

$$\frac{\Delta f_o}{f_o} \approx -\frac{C_3(G_2 + G_1)}{G_5(C_4 + C_3)} \frac{f_o}{2Q_o} \frac{1}{\text{GBP}} = -\frac{C_4 + C_3}{C_4} \frac{Q_o f_o}{2} \frac{1}{\text{GBP}} \quad (9)$$

$$\frac{\Delta Q_o}{Q_o} \approx \frac{C_3(G_2 + G_1)}{G_5(C_4 + C_3)} \frac{f_o}{2Q_o} \frac{1}{\text{GBP}} = \frac{C_4 + C_3}{C_4} \frac{Q_o f_o}{2} \frac{1}{\text{GBP}} \quad (10)$$

From Eqns. 9 and 10, we see that the errors grow as Q_o increases or f_o increases. When $C_4 = C_3$, then the error magnitude is $Q_o f_o / \text{GBP}$. So for this selection of components, we can have an error of 1% or less whenever $Q_o f_o < \text{GBP}/100$. This type of result can be found in all active filters and it is common to see a $Q_o f_o$ product specification in an active filter data sheet [10].

A closer examination of Eqns. 9 and 10 also suggests that picking $C_4 = C_3$ is not the best choice to reduce errors due to op-amp GBP. If we pick $C_4 \gg C_3$, then the error we found in our bandpass filter would drop by about half.

Developing a new design procedure using Eqns.1-4 and the above :

- 1) Pick C_3 to be a standard value
- 2) Pick $C_4 = 10 C_3$
- 3) From Eqn. 4, $R_5 = 11 Q_o / (10 \omega_o C_3)$
- 4) From Eqn. 3, $R_1 = Q_o / (10 H_{bp} \omega_o C_3)$
- 5) From Eqn. 1, $R_2 = Q_o / [(11 Q_o^2 - 10 H_{bp}) \omega_o C_3]$

Note that the last step implies that $11 Q_o^2 > 10 H_{bp}$. If this is not true then develop a new design procedure with $C_4 \neq 10 C_3$.

Redesigning for again $Q_o = 4.318$, $f_o = 20\text{k Hz}$ and $H_{bp} = 4$ then

- 1) Pick $C_3 = 390\text{p F}$
- 2) Pick $C_4 = 10 C_3 = 3900\text{p F}$
- 3) $R_5 = 11 Q_o / (10 \omega_o C_3) = 96.917\text{k } \Omega$ (Pick 100k Ω)
- 4) $R_1 = Q_o / (10 H_{bp} \omega_o C_3) = 2.2027\text{k } \Omega$ (Pick 2.2k Ω)
- 5) $R_2 = Q_o / [(11 Q_o^2 - 10 H_{bp}) \omega_o C_3] = 533.67 \Omega$ (Pick 510 Ω)

Using Eqn. 1-3, we can re-calculate the filter parameters with the nearest standard components or we could just re-run Symbolic SPICE. The output of **new_mfb_bp_std.cir** is listed in Table 9.

Table 9. Symbolic SPICE input file **new_mfb_bp_std.cir**

```
MFB BAND-PASS FILTER
V1 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XOA 0 4 2
.END
```

The output file **mfb_bp_std.fun** listed in Table 10 contains the filter parameters solved for by Symbolic SPICE.

Table 10. Symbolic SPICE Output File **new_mfb_bp.fun**

```
MFB BAND-PASS FILTER

SECOND ORDER FILTER PARAMETERS:

Qo is:


$$\frac{\text{SQRT}\{( + C4 * C3) * ( + G2 + G1)\}}{( + C4 + C3) * \text{SQRT}\{ + G5\}}$$


= 4.46781

Wo**2 is:


$$\frac{( + G5) * ( + G2 + G1)}{( + C4 * C3)}$$


fo = 20056Hz

*****

There exists a BAND PASS filter for : v2

BAND PASS GAIN (Hbp) is:


$$\frac{( - C3 * G1)}{( + G5) * ( + C4 + C3)}$$


= -4.13223

*****
```

Table 11 lists an edited version of the file **new_mfb_bp_std.cir** where XOA was replaced with XNOA.

Table 11. Symbolic SPICE input file **new_mfb_bp_noa.cir**

```
MFB BAND-PASS FILTER
V1 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XNOA 0 4 2
.END
```

Suppose we evaluate this filter using the NE5534 which has a gain-bandwidth-product of about 10M Hz. Running the input file shown in Table 11, the following are the prompts. The user responses are shown in bold:

Symbolic SPICE - Circuit Analyzer and Approximator
Demo Version 3.1
(C) Copyright 2010 by Willow Electronics, Inc.

```
INPUT FILE NAME [.cir] : new_mfb_bp_noa
Numerical evaluation of the results? (y/n) : y
Discard terms if their magnitude falls below a threshold? (y/n) : n
ERROR ANALYSIS FILE NAME [new_mfb_bp_noa.err] : (hit enter)
What is the GBP (in Hz) of XNOA? (y/n) : 10meg
```

The output file **new_mfb_bp_noa.err** listed in Table 12 is the analysis done by Symbolic SPICE on the shifting of poles in the characteristic equation.

Table 12. Symbolic SPICE Output File **mfb_bp_noa.err**

```
MFB BAND-PASS FILTER

*****
* Qo and fo *
*****

Qo is:

Sqrt{( + C4*C3)*( + G2 + G1)}
-----
( + C4 + C3)*Sqrt{ + G5}

= 4.46781

(2*PI*fo)**2 is:

( + G5)*( + G2 + G1)
-----
( + C4*C3)

fo = 20056 Hz
```

```

*****
* DQo/Qo = (D2-D0) foQo - Dfo/fo *
*****

D2-D0:      where k{i} = 1/GBP{i}

The numerator is:

+0 TIMES

+ 1

The denominator is:

( + G5*C4*C3)*( + G2 + G1)

*****
* Dfo/fo = (D2-D1) fo/(2Qo) *
*****

D2-D1 :      where k{i} = 1/GBP{i}

The numerator is:

-C3*k TIMES

+ G2 + G1

The denominator is:

( + G5)*( + C4 + C3)

*****
*      NUMERICAL EVALUATION      *
*****

Dfo/fo = -0.00492835

DQo/Qo = 0.00492835

```

What these results mean numerically is that there is a *decrease* in the center frequency f_o of approximately 0.492835 % and an *increase* in the filter Q_o of approximately 0.492835 %. In other words, the effect of having a 10 M Hz GBP is to shift f_o from 20,056 to an actual value of $20,056 (1 - 0.00492835) = 19,957$ Hz and to shift Q_o from 4.46781 to an actual value of $4.46781 (1 + 0.00492835) = 4.4898$.

The shifting in f_o and Q_o shifts the bandwidth which is f_o/Q_o . The bandwidth is shifted from $20,056 / 4.46781 = 4,489$ Hz to an actual value of $19,957 / 4.4898 = 4,445$ Hz.

When $C_4 = 10 C_3$, then the error magnitude is $11Q_o f_o / (20 \text{ GBP})$. So for this selection of components, we can have an error of 1% or less whenever $Q_o f_o < \text{GBP}/55$. This is an improvement of nearly two over the usual selection of $C_4 = C_3$.

Let's verify the Symbolic SPICE results with PSpice using an ideal op-amp in the first simulation and a one-pole model for the op-amp in the second. The input file is listed in Table 13 where $A_o = 100k$, $f_o = 100$ Hz and the GBP = 10M Hz. The output is graphed in Fig. 4.

Table 13. PSpice input file `new_bandpass.cir`

```
BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XOA 0 4 2 IDEAL
.SUBCKT IDEAL 1 2 3
RI 1 2 1000MEG
EA 3 0 1 2 1000MEG
.ENDS IDEAL
.AC DEC 20000 2K 200K
.PROBE
.END

BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XNOA 0 4 2 NE5534
.SUBCKT NE5534 1 2 3
EA 3 0 LAPLACE {V(1,2)} =
+ {(100K)*(6.283*100)/(S + (6.283*100))}
.ENDS NE5534
.AC DEC 20000 2K 200K
.PROBE
.END
```

From Fig. 4, for the ideal op-amp the peak frequency is 20.055k Hz, the peak gain is 4.1322 and the -3 dB bandwidth is 22.425k - 17.937k = 4.4874k Hz. Thus

$$f_o = 20.055\text{k Hz} \quad H_o = 4.1322 \quad Q_o = 20.055\text{k} / 4.4874\text{k} = 4.4692$$

and this is virtually identical to Symbolic SPICE.

From Fig. 4, for a non-ideal op-amp the peak frequency is 19.959k Hz, the peak gain is 4.1309 and the -3 dB bandwidth is 22.302k - 17.937k = 4.3647k Hz. Thus

$$\hat{f}_o = 19.959\text{k Hz} \quad \hat{H}_o = 4.1309 \quad \hat{Q}_o = 19.959\text{k} / 4.3647\text{k} = 4.573$$

Comparing to Symbolic SPICE, this is a difference of -0.01% for f_o and 1.85% for Q_o .

It is worth noting that the Symbolic SPICE error approximations are most accurate when the errors Symbolic SPICE calculates are in the 1% range or less. Although accuracy begins to depart around 10%, the terms dominating the error are still the same. This is helpful in making design decisions.




Figure 4. Redesigned MFB band-pass filter

I) Conclusion

Symbolic SPICE can be used to analyze ideal op-amp circuits to find transfer functions. It can recognize second order transfer functions in the Laplace operator s . Symbolic SPICE can determine the filter type such as band-pass and then solve for the filter parameters Q_o , ω_o and H_{bp} . It can also estimate the effects of op-amp's gain-bandwidth-product on the filter parameters, ω_o and Q_o . Lastly this application note showed that is possible to almost double the usable bandwidth of the MFB band-pass active filter by proper selection of capacitor values.

J) References

- [1] J. Graeme, G. Tobey and L. Huelsman, *Operational Amplifiers: Design and Applications* McGraw-Hill, pp. 292-293, 1972.
- [2] G. M. Wierzba, *ECE 402: Application of Analog ICs Class Notes*, Ch.2 , pp. 44-46. This e-book is available at <http://stores.lulu.com/willowpublishing>
- [3] B. Carter, "An Audio Circuit Collection, Part 3," <http://www.ti.com>
- [4] Elliott Sound Productions, Projects 63 and 64, <http://sound.au.com>

- [5] G. M. Wierzba, *ECE 402: Application of Analog ICs Class Notes*, Ch.2 , pp. 46-53.
- [6] G. M. Wierzba, *Symbolic SPICE User Manual*. This e-book is available at <http://stores.lulu.com/willowpublishing>
- [7] D. Bohn, "Constant-Q Graphic Equalizers," *J. Audio Eng Soc.*, Vol. 34, No. 9, Setp. 1986.
- [8] *Audio Handbook*, National Semiconductor, pp. 2.56-2.59, 1976.
- [9] G. Wilson, Y. Bedri and P. Bowron, "RC-Active Networks with Reduced Sensitivity to Amplifier Gain-Bandwidth-Product," *IEEE Transactions on Circuits and Systems*, CAS-21, pp. 618-626, September 1974.
- [10] "UAF42 - Universal Active Filter," <http://www.ti.com>